
Alarmist extends Erlang's Alarm
Handler to manage, subscribe to,
and create synthetic alarms. Code that sets and clears alarms need only depend
on :alarm_handler. Alarmist is for the side that has to deal with alarms.

  
    
  
  What are alarms


Alarms are different from events. While events can convey any information, an
alarm conveys a boolean state. The alarm can either be set or clear. Code
should always be able to know the state of the alarm. With events, you either
get the event or not. There may be ways of obtaining the event when it's missed,
but with alarms there's an expectation that the alarm state is always
accessible.
Erlang's Alarm Handler allows code that sets alarms to include supplementary
information called AlarmDescription. This is purely informational. If an alarm
is set more than once, only the latest description is available. It is not
useful for differentiating alarms. For example, a network disconnected alarm
should incorporate the network interface name (eth0) into the AlarmId rather
than the AlarmDescription.

  
    
  
  When to use alarms


Alarms are one tool in the fault management toolbox. They give a name to
persistent conditions that are involved with non-local remediation to clear.
Persistent in this sense means that the alarm continues to exist until reported
otherwise. It is not transient. For example, a supervised GenServer that crashes
is a transient fault since its supervisor is going to restart it. An issue like
a remote server no longer being reachable is persistent. It may become reachable
in a few seconds or hours or more.
Non-local remediation means that the code that sets the alarm does so to either
help or get help from somewhere else like another library or a person. For
example, code that monitors a network connection could set an alarm when the
internet is unreachable so that UI code could show the issue to a nearby human.

  
    
  
  Naming alarms


Erlang's Alarm Handler allows AlarmId's to be any Erlang term. While super
flexible, it's also helpful to have a convention.
For Elixir code, name alarms as you would a module. If you have helper functions
for AlarmDescription data, then put those functions in a defmodule of the
same name as the alarm. This is optional, so there's no need to create an empty
module if you don't have helper functions.
For libraries, alarms are public API. There's no explicit place for alarms in
Hex documentation, so add them where you think best. The important parts are to
document the alarms name, when it's set and cleared, and the type and content of
the AlarmDescription data.
Erlang code should use Erlang conventions for naming modules.
Using 2-tuples for AlarmId's so that you can have generic alarms is not
supported by Alarmist, but probably will be added. I.e. {NetworkDown, "eth0"}

  
    
  
  Synthetic alarms


One of the major features of Alarmist is to support creating new alarms that
get set and cleared based on the status of other alarms. This simplifies alarm
handling code since it's often the case that you don't want to trigger a
remediation immediately or a remediation may only be useful if some combination
of alarms are set. Another advantage of creating synthetic alarms is that it
increases the visibility of when these more complex alarms trigger so that you
don't need to dig through the alarm handling code to know whether conditions
have met the remediation criteria or not.
As before, networking issues make good examples. Home and business networks have
some normal flakiness that doesn't require remediation. Sometimes just waiting a
bit resolves a situation. Code that detects a network outage can simply set an
alarm stating it is down. Alarmist provides primitives for creating a
synthetic alarm that doesn't get set unless the network is down longer than a
minimum amount of time. Alarmist can also raise an alarm if the network is
bouncing up and down a lot since that's also problematic, but in a way that the
minimum time criteria wouldn't detect.
To make synthetic alarms easy to create, Alarmist provides the defalarm
macro. The general form is to create a module with the name of the synthetic
alarm you're creating and then use defalarm to express the criteria for it
being set:
defmodule MyNewAlarm do
  use Alarmist.Definition

  defalarm do
    InterestingAlarm1 and InterestingAlarm2
  end
end
In this example, MyNewAlarm will be set only when both InterestingAlarm1 and
InterestingAlarm2 are set.
The following sections describe the operators available for synthetic alarms.

  
    
  
  Identity


Specifying an AlarmId by itself creates a new alarm whose state mirrors the
original one. In other words, it creates an alias and is useful for decoupling
the naming of alarms between projects.
defmodule IdenticalAlarm do
  use Alarmist.Definition

  defalarm do
    SomeOtherAlarmName
  end
end

  
    
  
  Debounce


The debounce/2 function specifies a minimum amount of time for another alarm
to be set before it is set. This can be used to delay remediation if there's a
chance that the alarm goes away on its own.
defmodule RealProblemAlarm do
  use Alarmist.Definition

  defalarm do
    # Set this module's alarm when FlakyAlarm has been set for at for 5 seconds
    debounce(FlakyAlarm, 5_000)
  end
end

  
    
  
  Hold


The hold/2 function specifies a minimum amount of time for the new alarm to be
set. For example, if an alarm triggers an indicator on a UI, then it may need to
stay on for a minimum duration. While the UI could have the timer, creating an
alarm lets other code or alarms change their behavior as well.
defmodule LongerAlarm do
  use Alarmist.Definition

  defalarm do
    # Set the alarm for at least 3 seconds whenever FlakyAlarm
    hold(FlakyAlarm, 3_000)
  end
end

  
    
  
  Intensity


The intensity/3 function sets an alarm when another has been set and cleared
too many times in a row. The metric is set/cleared x times in y milliseconds
similar to OTP's supervisor restart intensity parameters. It can be useful to
combine intensity/3 with hold/2 to create an alarm that disables a feature
for a short time when it flaps too much. Some people call this a penalty box.
defmodule IntensityThresholdAlarm do
  use Alarmist.Definition

  defalarm do
    # Set when raised and cleared >= 5 times in 3 seconds
    intensity(FlakyAlarm, 5, 3_000)
  end
end

  
    
  
  Boolean logic


Standard Elixir boolean operators like and, or, and not can be used to
combine and group multiple alarms. This is an easy way to create an alarm that
tracks exactly what you want.
defmodule IntensityThresholdAlarm do
  use Alarmist.Definition

  defalarm do
    (Alarm1 or Alarm2) and intensity(FlakyAlarm, 5, 10_000)
  end
end

  
    
  
  Example


The following example shows how to define an alarm that WiFi is unstable based
on a alarm that says when WiFi is down. This is a real life example of an
embedded device with an expensive backup cellular connection. WiFi can be flaky,
though, so you wouldn't want to turn on the cellular connection right when WiFi
goes down since that might be a hiccup.
The following code defines a synthetic alarm for unstable WiFi,
Demo.WiFiUnstable. The timeouts are short to make it easier to copy/paste into
an IEx prompt and manually run.
defmodule Demo.WiFiUnstable do
  @moduledoc """
  Alarm for when WiFi bounces too frequently
  """
  use Alarmist.Definition

  # WiFi must be down for at least 15 seconds or flapped 2 times in 60 seconds
  defalarm do
    debounce(Demo.WiFiDown, :timer.seconds(15)) or
      intensity(Demo.WiFiDown, 2, :timer.seconds(60))
  end
end

defmodule Demo do
  @moduledoc """
  Helpers for setting and clearing alarms
  """
  def wifi_down() do
    :alarm_handler.set_alarm({Demo.WiFiDown, nil})
  end

  def wifi_up() do
    :alarm_handler.clear_alarm(Demo.WiFiDown)
  end

  def wifi_flap() do
    wifi_down()
    wifi_up()
    wifi_down()
    wifi_up()
  end
end
Now that we have alarm logic and helpers defined the synthetic alarm needs to be
registered:
  # ... normally in an Application.start or other code that runs on init ...
  Alarmist.add_synthetic_alarm(Demo.WiFiUnstable)
Then subscribe for notifications:
  # ... normally in the GenServer with the remediation code...
  Alarmist.subscribe(Demo.WiFiUnstable)
Finally, we can exercise setting and clearing the alarm:
iex> Demo.wifi_flap
:ok
iex> flush
%Alarmist.Event{
  id: Demo.WiFiUnstable,
  state: :set,
  description: nil,
  timestamp: -576460712978320952,
  previous_state: :unknown,
  previous_timestamp: -576460751417398083
}
:ok
# Wait ~60 seconds
iex> flush
%Alarmist.Event{
  id: Demo.WiFiUnstable,
  state: :clear,
  timestamp: -576460652977733801,
  previous_state: :set,
  previous_timestamp: -576460712978320952
}

  
    
  
  License


Alarmist is licensed under the Apache License, Version 2.0.


  

    
Changelog
    

This project follows Semantic Versioning.

  
    
  
  v0.2.1 - 2025-03-27


	Updates	Fix serious issue with incorrect clearing of timers that affects synthetic
alarms that use timers such as Debounce, Intensity, and Hold. Timeouts could
be missed. Thanks to @x4lldux for reporting the issue.
	Improve compile-time checks for defalarm
	Various documentation improvements and an example
	Update licensing and copyright for REUSE compliance




  
    
  
  v0.2.0 - 2024-12-09


This is a backwards incompatible update. The following changes are needed:
	Replace all calls to Alarmist.current_alarms/0 with
Alarmist.get_alarm_ids/0. This is a hard deprecation.
	Update all message handling on Alarmist events to expect and use the
Alarmist.Event.t(). In most usage, this means matching on :id and
:state and it should simplify the handling functions.

	Updates	Simplify the API by completely abstracting away the internal implementation
with uses the PropertyTables library. This will allow for further internal
improvements without forcing breaking changes.
	Align the API for getting alarms with :alarm_handler. This added the
Alarmist.get_alarms/0 function.
	Remove a race condition involving the use of alarm descriptions.
Descriptions are sent to subscribers with the alarm status change
notification now.
	Use one timestamp value for all alarms that were set at initialization time.
This removes the ambiguity of whether an alarm changed a few milliseconds
after Alarmist start up or was one of the original alarms.
	Report when alarms are in an :unknown state when no information is
available. This is useful for the :previous_state field in alarm events.




  
    
  
  v0.1.3 - 2024-09-26


	Updates	Don't crash on non-atom Alarm IDs. Alarmist doesn't support these yet so
they're currently ignored.




  
    
  
  v0.1.2 - 2024-03-04


First public release

  
    
  
  v0.1.1 - 2024-02-29


	Updates	Delay swapping alarm handler until supervision tree started to fix possible
crash on startup




  
    
  
  v0.1.0 - 2024-01-19


Initial release

Alarm handler and more
Alarmist provides an :alarm_handler implementation that allows you to check
what alarms are currently active and subscribe to alarm status changes.
It also provides a DSL for defining alarms based on other alarms. See
Alarmist.Definition.

          @type alarm() :: {alarm_id(), alarm_description()}


      


Alarm information
Calls to :alarm_handler.set_alarm/1 pass an alarm identifier and
description as a 2-tuple. Alarmist stores the description of the most recent
call.
:alarm_handler.set_alarm/1 doesn't enforce the use of 2-tuples. Alarmist
normalizes non-2-tuple alarms so that they have empty descriptions.

Alarm description
This is optional supplemental information about the alarm. It could contain
more information about why it was set. Don't use it to differentiate between
alarms. Use the alarm ID for that.

          @type alarm_id() :: atom()


      


Alarm identifier
Alarm identifiers are the unique identifiers of each alarm that can be
set or cleared.
While SASL alarm identifiers can be anything, Alarmist imposes the restriction
that they all be atoms. It is highly recommended to use module names to
avoid naming collisions. Non-atom alarms are currently ignored by Alarmist.

          @type alarm_state() :: :set | :clear


      


Alarm state
Alarms are in the :set state after a call to :alarm_handler.set_alarm/1
and in the :clear state after a call to :alarm_handler.clear_alarm/1.
Redundant calls to :alarm_handler.set_alarm/1 update the alarm description
and redundant calls to :alarm_handler.clear_alarm/1 are ignored.

Add a rule-based alarm
After this call, Alarmist will watch for alarms to be set based on the
supplied rules and set or clear the specified alarm ID. The alarm ID
needs to be unique.

  



  
    
      
    
    
      add_synthetic_alarm(alarm_id, compiled_rules)



        
          
        

    

  


  

      

          @spec add_synthetic_alarm(alarm_id(), compiled_rules()) :: :ok


      


Manually add a rule-based alarm
Use this when not using defalarm.
After this call, Alarmist will watch for alarms to be set based on the
supplied rules and set or clear the specified alarm ID. The alarm ID
needs to be unique.

Return a list of all active alarm IDs

Return a list of all active alarms
This returns {id, description} tuples. Note that Alarmist normalizes
alarms that were not set as 2-tuples so this may not match calls to
:alarm_handler.set_alarm/1 exactly.

Manually add a rule-based alarm
Use this when not using defalarm.
After this call, Alarmist will watch for alarms to be set based on the
supplied rules and set or clear the specified alarm ID. The alarm ID
needs to be unique.

          @spec subscribe(alarm_id()) :: :ok


      


Subscribe to alarm status events
Events will be delivered to the calling process as:
%Alarmist.Event{
  id: TheAlarmId,
  state: :set,
  description: nil,
  timestamp: -576460712978320952,
  previous_state: :unknown,
  previous_timestamp: -576460751417398083
}

Unsubscribe the current process from the specified alarm :set and :clear events

Compiles synthetic alarm definitions into a list of rules

DSL for defining alarms
The general form of an alarm definition is:
defmodule MyAlarmModule do
  use Alarmist.Definition

  defalarm do
    AlarmId1 and AlarmId2
  end
end
See Alarmist.Ops for what operations can be included in defalarm block.

Synthetic alarm processing engine
This module is intended for users extending the Alarmist DSL.

	:cache - temporary cache for alarm status while processing rules
	:changed_alarm_id - list of alarm_ids that have changed values
	:timers - map of alarm_id to pending timer
	:actions_r - list of pending side effects in reverse (engine processing is side-effect free
by design so someone else has to do the dirty work)
	:states - optional state that can be kept on a per-alarm_id basis
	:lookup_fun - function for looking up alarm state

Create and add a synthetic alarm based on the rule specification
The synthetic alarm will be evaluated, so if the synthetic alarm ID already
has subscribers, they'll get notified if the alarm is set.
Raises on errors

Cache alarm state and record the change
IMPORTANT: Rules are evaluated on the next call to run/2 if there was a change.

Commit all side effects from previous operations
The caller needs to run all of the side effects before the next call to the engine
so state changes may be lost.

Remove all of the rules associated with the specified id

Struct sent to subscribers on property changes
	:id - which alarm
	:state - :set or :clear
	:description - alarm description or nil when the alarm has been cleared
	:timestamp - the timestamp (System.monotonic_time/0) when the changed happened
	:previous_state - the previous value (:unknown if no previous information)
	:previous_timestamp - the timestamp when the property changed to
:previous_state. Use this to calculate how long the property was the
previous state.

Derivative alarm generation operations

      


      
        Summary


  
    Functions
  


    
      
        copy(engine, list)

      


        Replicate an alarm status



    


    
      
        debounce(engine, list)

          @spec copy(Alarmist.Engine.t(), [Alarmist.alarm_id()]) :: Alarmist.Engine.t()


      


Replicate an alarm status
This is useful for aliasing alarm names. For example, if one library sets and
clears an alarm ID that's in its namespace, but another library wants to
listen on changes to an alarm ID in its namespace, a copy rule can glue them
together.
Example:
defmodule NewAlarm do
  use Alarmist.Definition

  defalarm do
    OriginalAlarm
  end
end

          @spec debounce(Alarmist.Engine.t(), [Alarmist.alarm_id(), ...]) :: Alarmist.Engine.t()


      


Set an alarm when the input has been set for a specified duration
This rule removes transient alarms from triggering remediation unnecessarily.
This is useful when remediation is expensive or service impacting and the
input alarm is somewhat glitchy.
Alarmist already provides some debouncing since alarms that get set and
cleared in one alarm processing pass are ignored already. This is unreliable,
though, and a debounce rule establishes a duration.
An example of when debouncing is useful is to delay remediation of higher
level alarms like being disconnected from a backend server. There are many
reasons that a TCP connection could be interrupted and client code probably
has some retry logic in it already to reestablish the connection. In this
case, it might be good to delay switching to an offline mode for a little bit
in the hopes that the problem will naturally go away.
Example:
defmodule NewAlarm do
  use Alarmist.Definition

  defalarm do
    debounce(Alarm1, 1_000)
  end
end

          @spec hold(Alarmist.Engine.t(), [Alarmist.alarm_id(), ...]) :: Alarmist.Engine.t()


      


Keep an alarm set for a guaranteed amount of time
This sets an alarm for at least timeout milliseconds after it is set. Each
time the alarm is set, the timer is restarted.
Hold is useful for types of remediation that are time based. I.e., handling
an alarm means turning something off for a while since turning that feature
back on when the alarm gets cleared would likely just result in the alarm
being set again. Managing the timeout period via alarms rather than
programmatically lets you manually clear the alarm if you'd like that feature
enabled again immediately like if you're debugging.
Example:
defmodule NewAlarm do
  use Alarmist.Definition

  defalarm do
    hold(Alarm1, 1_000)
  end
end

          @spec intensity(Alarmist.Engine.t(), [Alarmist.alarm_id(), ...]) ::
  Alarmist.Engine.t()


      


Sets an alarm when the input alarm has been set and cleared too many times
This type of rule catches flapping alarms where it's desirable to take some
kind of remediation when they trigger too many times in a row. Intensity is
measured as count set/clears in duration milliseconds. This is the same
as supervision restart intensity thresholds.
An example of an intensity-based alarm is to handle the case when multiple
network connections are available, but one that should be good is flakey.
This happens if a device has both a cellular and a WiFi connection. Normally
the WiFi connection is preferred, but if it keeps going up and down, it may
be desirable to raise an alarm. That alarm could disable WiFi for a while.
Combine this with hold/2 to manage the duration that WiFi is off.
Example:
defmodule NewAlarm do
  use Alarmist.Definition

  defalarm do
    intensity(Alarm1, 3, 60_000)
  end
end

          @spec logical_and(Alarmist.Engine.t(), [Alarmist.alarm_id()]) :: Alarmist.Engine.t()


      


Set an alarm when all of the input alarms are set
This is useful when remediation is only useful when a lot of things go wrong.
For example, if a device has more than one way of accomplishing a task, there
could be a specific remediation when one way stops working. However, if every
way is broken, the device could trigger a more significant remediation.
Example:
defmodule NewAlarm do
  use Alarmist.Definition

  defalarm do
    Alarm1 and Alarm2
  end
end

          @spec logical_not(Alarmist.Engine.t(), [Alarmist.alarm_id()]) :: Alarmist.Engine.t()


      


Set an alarm when the input alarm is cleared
This is useful for "proof-of-life" alarms where the presence of an alarm is a
good thing.
Example:
defmodule NewAlarm do
  use Alarmist.Definition

  defalarm do
    not OriginalAlarm
  end
end

          @spec logical_or(Alarmist.Engine.t(), [Alarmist.alarm_id()]) :: Alarmist.Engine.t()


      


Set an alarm when one or more input alarms get set
This is useful for triggering a generic remediation. An example of this for
setting an alarm that indicates that the device is "unhealthy" and needs to
reboot. There are usually many disastrous alarms that when raised really have
no great remediation other than reboot. This allows a handler to register for
a combined alarm so that it's decoupled from the alarms that trigger it.
Example:
defmodule NewAlarm do
  use Alarmist.Definition

  defalarm do
    Alarm1 or Alarm2
  end
end

